USE YOUR TRS-80 COLOR COMPUTER AS A STORAGE OSCILLOSCOPE



Fig. 1

With an analog-to-digital converter on your TRS-80 Color Computer, you have the makings of a low-cost storage oscilloscope

By Forrest M. Mims, III

Of the many kinds of electronic test instruments, the oscilloscope is by far the most important. Only the oscilloscope provides a graph-like electronic picture showing how a voltage (the *signal*) varies with time. You might say the oscilloscope combines the functions of a voltmeter and frequency counter into a single but more powerful instrument. Borrowing from the old adage, you might also say a single oscilloscope trace is worth a thousand data points.

In the simplest oscilloscopes, a thin beam of electrons is repeatedly swept across the phosphor screen of a cathode ray tube (CRT) much like those in television sets. The beam can be deflected vertically by a signal applied to a sensitive amplifier whose ouput is connected to electrodes within, or an electromagnetic coil outside, the CRT. This provides a brightly illuminated graph that neatly displays the time-dependent variations in the amplitude of a signal. Simply by speeding up the time required for the beam to sweep across its screen, an oscilloscope can provide a look at signals whose amplitude fluctuates many millions or even billions of times in a single second.

Often an oscilloscope user would like to be able to "freeze" the trace on an oscilloscope screen for more detailed study. This is especially true when the sweep speed is very slow or when a signal with a very brief duration (such as that produced by a switching logic gate) is being monitored.

One way to preserve a trace is to photograph the oscilloscope's screen. Another is to use a sophisticated instrument called a *storage oscilloscope*. Unfortunately, storage oscilloscopes are quite costly.

A little more than a year ago, a small company called Northwest Instrument Systems (PO Box 1309, Beaverton, OR 97075) revolutionized oscilloscope technology when it introduced a sophisticated computer peripheral that brought the capabilities of a storage oscilloscope to anyone with an Apple II computer and $995. Dubbed the Model 85 aScope™, the device is a module containing a fast analog-to-digital converter. When plugged into an Apple II, the aScope provides the capabilities of a laboratory storage oscilloscope costing as much as $10,000.

Fig. 2. Three circuits whose outputs can be monitored.



The aScope is supplied with appropriate software (on disk) to place labels and a graticule on the screen of the computer's monitor. The software also permits traces to be saved on disk and printed on paper.

The aScope is merely the first of an expected flood of peripherals that will allow personal computers to function as many kinds of sophisticated, programmable test instruments. Concerned about this new development, traditional makers of oscilloscopes and other test instruments have already begun to develop new systems designed around personal computers.

If you own or have access to a personal computer equipped with an analog-to-digital converter, you can develop your *own* test instruments. You can even convert your system into a slow-scan, storage oscilloscope. In the remainder of this article we will explain how.

If you own a personal computer with an analog-to-digital converter, you have the essential ingredients for a slow-scan oscilloscope. And if you have access to an x-y plotter, you can transform traces displayed on the computer's display into publication-quality images.

Best of all, you can assemble a computer-simulated oscilloscope from surprisingly economical components. While many different computers can be used, I have had excellent results with Radio Shack's Color Computer. Using a system equipped with Extended Color BASIC, I developed software to transform the computer's display into a very useful large-screen, slow-scan oscilloscope with both single-sweep (storage mode) and continuous-sweep operation. The software includes an optional graticule complete with tick marks, and the ability to store traces on a floppy disk for later retrieval and display. A second software package directs an optional plotter to draw a high-resolution rendition of a trace stored on a disk.

Getting Started. If you've ever tried to use a standard oscilloscope to observe a slowly changing signal, you can readily appreciate the advantage of a storage scope that freezes the display for later viewing and interpretation. The only other way to preserve the trace is to photograph the display.

A simple block diagram showing how a personal computer can be configured as a slow-scan storage oscilloscope is shown in Fig. 1. While any of a number of computers can be used, excellent results can be attained with a 32K Color Computer with Extended Color BASIC.

The original machine has been replaced by the 16K Extended Color Computer 2 and the 64K Extended Color Computer. Like the original product, these new machines are equipped with two analog joystick ports for a total of four A/D converter inputs. The software given here, although developed for the original Color Computer, should run on both new machines.

Operation of the Color Computer joystick ports is described in detail in "The Electronic's Scientist" column elsewhere in this magazine. Briefly, when the joystick feature is selected, the computer generates a decimal number directly proportional to the voltage at the selected joystick port. These numbers ranges from 0 to 63 (6-bit accuracy) when the voltage varies from about 0.3 to 4.75 volts.



Fig. 3. Charge-discharge cycle for circuit shown in Fig. 2A.



Fig. 4. Two spikes produced when flashlight is blinked twice.



Fig. 5. Slow scan of flashlight across circuit in Fig. 2C.

Normally the voltage is derived from the two potentiometers in each joystick. Each potentiometer is connected as a voltage divider across a 5-volt supply from the computer with the wiper terminal providing a variable output voltage as the potentiometer shaft is rotated.

Of course any other voltage source can be connected to the joystick ports if its excursions fall within a 0-to-5-volt window. In any case, the digitized representation of the voltage can be summoned by the simple command JOYSTK(*n*) where *n* is the port number (0-3).

Knowing this and taking advantage of the Color Computer's powerful graphics capabilities, it is easy to develop a rudimentary oscilloscope program. Here's one possibility:

```
10 PCLS : PMODE 4 , 1 : SCREEN
   1 , 0
20 FOR X=0 TO 255
30 Y=JOYSTK ( 0 )
40 PSET ( X , Y )
50 NEXT X
60 GOTO 10
```

This very simple program transforms the screen of the monitor into a black square surrounded by a green border. A moving green dot traces a solid line from left to right across the black square.

If the wiper of the potentiometer representing JOYSTK(0) (horizontal, right joystick) is rotated, the green dot will move up (or down), leaving behind a trail of dots. When the dot reaches the right side of the square, the line and dot pattern it has traced is instantly erased and a new sweep is begun.

The simple oscilloscope program can be converted to emulate a storage scope merely by changing the last line to:

```
60 GOTO 60
```

The green dot will now form a single trace which will be displayed on the monitor until the BREAK key is pressed.

Other enhancements can be easily added to the basic program. For example, the following BASIC line will add a trigger feature to intitate a trace *only* when the incoming voltage exceeds 0.3 volt:

```
15 IF JOYSTK ( 0 ) > 0 THEN 20
   ELSE 15
```

The trigger feature can be further enhanced by changing 0 [not JOYSTK(0)] to a variable. Many other features can also be added. For example, a graticule can be drawn on the monitor's screen. And, traces can be saved on disk for later retrieval and viewing.

Comscope. The simple program just discussed has been enlarged into a menu-driven computerized oscilloscope program called Comscope shown in Listing 1. While the joystick and graphics commands in Compscope are unique to the Extended Color BASIC of the Color Computer, the program can be used as a guide to develop similar software for other computers.

When entered and run, the monitor displays Comscope's menu:

```
COLOR COMPUTER
OSCILLOSCOPE
A.SINGLE SWEEP STORAGE
     SCOPE.
B.CONTINUOUS SWEEP SCOPE.
C.RETRIEVE TRACE FROM
     DISK.
FOR GRATICULE ADD G TO
     SELECTION
PRESS M TO RETURN TO MAIN
     MENU
SELECTION?
```

The menu provides a total of six options. If a G is appended to the selection, the graticule subroutine (line 2000-2160) is executed. The graticule provides a grid of 16 horizontal and 12 vertical divisions. The center horizontal and vertical lines are given short tick marks.

If single or continuous sweep modes are selected, the sweep subroutine (lines 4000-4700) is executed. In both cases, the program first asks for the desired sweep time in seconds-per-division (0.35 second minimum) and assigns it to variable SD (see lines 340-360).

The program subtracts 0.34 second from the time entered. The 0.34 s is the approximate time required for the program to execute the steps that retrieve joystick values and draw the trace on the monitor (lines 4300-4350). The result after the subtraction is multiplied by 460, the number that, when used in a one line timer loop (FOR T=0 TO 460:NEXT T), produces a delay of 1 second.

If the trace is *not* to be saved on disk, the product is then divided by 16, the number of pixels in a single graticule division, and designated SW (line 360). If the trace is to be saved on disk, the product is divided by 4. This is because traces to be saved on disk must be made at 1/4 the resolution of ordinary traces to prevent the disk buffer from filling before all or at least most of the trace has been completed.

High resolution traces *can* be saved, but since the disk drive will turn on and momentarily interrupt program execution several times during a single trace, the resultant traces will contain undesirable artifacts. This is particularly true when fast sweep speeds are selected and when the input voltage varies rapidly.

The processed seconds-per-division variable (SW) is held until needed by the sweep subroutine. It is then used in the timer loop at line 4340.

In the simple demonstration program given earlier, the trace was formed by a stream of dots. When superimposed on a graticule, however, the dots lack the visibility a solid line would provide.

For this reason the sweep subroutine forms a continuous line trace. Line 4100 establishes a starting point for the LINE command. Line 4330 then extends a line from the starting point to the first pair of coordinates designated by a FOR . . . NEXT loop (x values) and the data retrieved from the disk data file (y values).

Incidentally, line 4200 in the sweep subroutine provides a trigger that prevents the sweep from beginning until the voltage at the joystick port exceeds about 0.3 volt. This feature can be omitted by dropping line 4200. Or you can alter the condition so that the sweep begins at any joystick value you select (0-63).

A particularly important option in the single-sweep storage scope mode is whether or not you wish to save a trace on disk. Comscope assigns your answer (Y or N) to a string variable called D$ and at various points in the program queries D$ to determine whether or not data is to be saved on disk.

At line 370, a disk data file assigned a name you specify (see N$, lines 330 and 370) is opened if D$ is Y. At line 4320 in

(Continued on page 98)



Fig. 6. How to connect a plotter to the Color Computer.



Fig. 7. Plot of joystick output.



Fig. 8. Plot same as Fig. 3.



Fig. 9. Plot same as Fig. 4.



Fig. 10. Plot same as Fig. 5.

Storage Oscillosope

Prior to running Scopeplt, the plotter's scaling points should be manually entered into the plotter. This is accomplished from the plotter's control panel by first directing the pen to the desired upper *left* corner of the plot and entering Pl. Next, move the pen to the desired lower *right* corner of the plot and enter P2. Be sure to leave room for the label which will be drawn under the plot.

Next, select the pens you wish the plotter to use. A fine tip (0.3 mm) black pen is excellent for the graticule or border (right stall) while a broad tip (0.7 mm) red, green or blue pen provides a highly visible trace (left stall).

Make sure the plotter is placed so the paper has an unimpeded path as it moves back and forth across the bed. Then place a sheet of paper in the plotter. (White typing bond is okay, but plotter paper provides publication-quality results.)

Now run Scopeplt. The program will ask you to enter the name of the trace you want to plot. (You'll get an NE error if the data file for the trace is not on the disk inserted in the drive.) The program will then ask if you want a graticule. If not, it will ask if you want tick marks inserted around the rectangle which will surround the plot.

After you have responded to the program, Scopeplt will take over and rapidly commence drawing the plot. After completing the graticule (or border) and trace, it will label the plot and insert the calibration factors. The program uses a fixed value of 0.4 volt per vertical division, with the value for the sweep speed retrieved from the disk file.

Be sure to keep your fingers away from the fast moving pen! Depending upon where you set P1 and P2, you can enter new locations for P1 and P2 and make additional plots on the same sheet.

A sample test plot showing voltage fluctuations generated by one axis of an analog (potentiometer voltage divider) joystick is shown in Fig. 7. The data for this oscillogram, which illustrates the border with tick marks option, was retrieved from a disk data file called JOYSTK-1.

Figures 8, 9, and 10 are plotter-generated oscillograms corresponding to the computer monitor photographs in, respectively, Figs. 3, 4 and 5. All three oscillograms illustrate the graticule option.

Various combinations of pen colors and widths can be used to dress up the oscillograms. You can plot an oscillogram anywhere you please on a sheet of paper simply by setting new P1 and P2 locations. You may wish to make light pencil marks for the new locations. Enter them after first moving a pen directly over each mark. Remember that you'll have to reenter the P1 and P2 points if you turn off the plotter.

Modifying Scopeplt. Experienced HPGL programmers will have little trouble revising Scopeplt to include additional features. With a little practice, even novice HPGL programmers can soon learn to expand the program.

The most obvious revision would be a menu of options to permit such a frill as keyboard control of the location, size, and orientation of the oscillograms. Other options might include user-selectable label size and graticule configuration. A text feature could also be included.

The programs can be revised to drive other computers and plotters

Incidentally, normally P1 and P2 are set, respectively, at the upper *left* and lower *right* corners of the paper. Scopeplt alters this arrangement in order to match the Color Computer coordinate system. Therefore, the labels are *reverse* printed so they will apppear normal (see line 910). Keep this in mind if you revise the program.

Going Further. Both Comscope and Scopeplt illustrate the use of moderately priced computer hardware to provide capabilities previously unavailable to engineers, technicians, and experimenters unable to afford expensive storage oscilloscopes and data loggers. Remember that, while these programs were developed for the Color Computer and HP-7470, they can both be revised to drive *other* computers and plotters. ◇

Sharp PC-5000

(Continued from page 63)

tem, no special procedure is required for using the bubble memory. Bubble cartridges are small enough to tuck into a shirt pocket and are quite a bit more rugged than floppy disks and data cassettes.

As I see it, the only drawbacks to the bubble-memory system are its limited 128K storage capacity and the relatively high price of additional cartridges. However, with a combination of disks and bubble memory, the one cartridge supplied with the computer should be sufficient for most users.

Closing Comment. Sharp has done a great job in designing the PC-5000. My main regret is that I didn't have a production model, complete with software and all options, to test. The fact that the computer uses MS-DOS as its operating system should mean that there will be many fine software packages available for the PC-5000 in a short time. If Sharp supports and services the computer, the PC-5000 should be a real winner. ◇

Storage Oscilloscope

The program combines instructions for the computer's video monitor *and* the plotter so that both form graticules (if so instructed by the user) before plotting the retrieved trace.

The program probably seems very complicated if you're not familiar with HPGL. Actually, HPGL is very easy to use, as you can readily see by referring to the keystone of the program, lines 500-800. These five short lines form a loop that plots the saved trace a point at a time on both the computer's monitor and the plotter. Here's what happens:

Line 500 establishes a loop of 256/4 events, 256 being the number of pixels across the screen of the monitor. The loop is stepped through each fourth pixel since the disk data is stored at 1/4 resolution.

Line 600 retrieves y-axis data (voltage) from the disk.

Line 700 moves the plotter pen to the appropriate position (PA or *P*lot *A*bsolute) and places the pen tip on the paper (PD or *P*en *D*own).

Line 800 repeats the cycle started at line 500.

As the loop executes, a stream of dots is "painted"on the video monitor, while simultaneously, the plotter pen draws a continuous line plot identical to the displayed trace.

While Scopeplt, like all graphics programs, is very meticulous in its detail, it's actually a relatively simple program. Look, for instance, at line 90. This line uses the plotter's SC (*SC*ale) instruction to establish a grid of 192 x 256 plotter units. This gives the plotter the same resolution as the Color Computer and allows the *x* and *y* values used for the computer to be used by the plotter.

The HPGL commands between lines 230 and 390 draw the graticule and insert tick marks. The HPGL commands between lines 3000 and 4800 draw a simulated CRT screen *without* a graticule and with or without border tick marks if either of these options is selected.

You may be wondering about the sprinkling of seemingly functionless time-wasting loops such as those in lines 310, 400, 3000, 3900, 4300, and 4700. These are included to prevent the buffer in the HP-7470A plotter from being filled by too rapid a stream of incoming instructions from the Color Computer.

To reduce execution time, experiment with the values in these timer loops.

How to Use Scopeplt. An HP-7470A (RS-232 version) plotter is connected to the serial port of a Color Computer as shown in Fig.6. While this arrangement does not permit the plotter to send data *to* the computer, it is satisfactory for most common plotting applications. Other computers offer better interfacing via "handshake" signals.

For more information on advertised products, equipment tested, etc., circle appropriate number on postpaid Free Information Card.

LISTING 1—COMSCOPE

```
10 'COMSCOPE
20 'COPYRIGHT 1984 BY FORREST M. MIMS III
30 CLS
50 PRINT @ 34, "COLOR COMPUTER OSCILLOSCOPE"
100 PRINT @ 96, "A.SINGLE SWEEP STORAGE SCOPE."
110 PRINT @ 160, "B.CONTINUOUS SWEEP SCOPE."
120 PRINT @ 224, "C.RETRIEVE TRACE FROM DISK."
150 PRINT @ 320, "FOR GRATICULE ADD G TO SELECTION"
160 PRINT @ 384, "PRESS M TO RETURN TO MAIN MENU"
165 PRINT "":PRINT ""
170 INPUT "SELECTION";SE$
200 IF SE$="B" OR SE$="BG" THEN 340
210 IF SE$="C" OR SE$="CG" THEN 1000
300 CLS
310 INPUT "SAVE TRACE ON DISK (Y OR N)";D$
320 IF D$="Y" OR D$="N" THEN 330 ELSE 300
330 IF D$="Y" THEN INPUT "NAME OF TRACE";N$
340 INPUT "SECONDS PER DIVISION (0.35 MINIMUM)";SD
350 IF D$="Y" THEN M=4 ELSE M=16
360 SW=((SD-.34)*460)/M
370 IF D$="Y" THEN OPEN "O",#1,N$
380 PMODE 4,1:PCLS:SCREEN 1,0
400 IF SE$="AG" OR SE$="BG" THEN GOSUB 2000
450 GOSUB 4000
500 IF SE$="B" OR SE$="BG" THEN 380
1000 CLS
1020 INPUT "NAME OF COMSCOPE TRACE ON DISK";N$
1030 OPEN "I",#1,N$
1100 PMODE 4,1:PCLS:SCREEN 1,0
1200 IF SE$="CG" THEN GOSUB 2000
1300 LINE (0,191)-(0,191),PSET
1310 FOR X=0 TO 255 STEP 4
1320 INPUT #1,Y
1330 LINE -(X,Y),PSET
1340 NEXT X
1350 CLOSE #1
1400 IF INKEY$="M" THEN 30 ELSE 1400
2000 'DRAW GRATICULE
2010 FOR H=0 TO 255 STEP 16
2020 LINE (H,0)-(H,191),PSET
2030 NEXT H
2040 FOR T=0 TO 191 STEP 4
2050 PSET (127,T):PSET (129,T)
2060 NEXT T
2100 FOR V=16 TO 191 STEP 16
2110 LINE (0,V)-(255,V),PSET
2120 NEXT V
2130 FOR T=0 TO 255 STEP 4
2140 PSET (T,95):PSET (T,97)
2150 NEXT T
2160 RETURN
4000 IF D$="Y" THEN Q=4 ELSE Q=1
4100 LINE (0,191)-(0,191),PSET
4200 IF JOYSTK(0) > 0 THEN 4300 ELSE 4200
4300 FOR X=0 TO 255 STEP Q
4305 IF INKEY$="M" THEN 30
4310 Y=189-3*JOYSTK(0)
4320 IF D$="Y" THEN WRITE #1,Y
4330 LINE -(X,Y),PSET
4340 FOR T=1 TO SW:NEXT T
4350 NEXT X
4500 IF D$="Y" THEN WRITE #1, SD
4600 IF SE$="A" OR SE$="AG"
     THEN CLOSE #1 ELSE RETURN
4700 IF INKEY$="M" THEN 30 ELSE 4700
```

LISTING 2—SCOPEPLT

```
10 'SCOPEPLT
15 'COPYRIGHT 1984 BY FORREST M. MIMS III
20 PRINT #-2, "SP0;":CLS
25 PRINT @ 34, "COLOR COMPUTER OSCILLOSCOPE"
30 PRINT @ 69, "HP7470 PLOTTER DRIVER"
35 PRINT @ 288, "PRESS ANY KEY TO START..."
40 IF INKEY$="" THEN 40 ELSE CLS
45 PRINT "SCOPE DATA FILE MUST BE ON DISK."
47 PRINT "PRESS R TO RESTART PROGRAM.":PRINT ""
50 INPUT "SCOPE DATA FILE NAME";N$
60 INPUT "DO YOU WANT A GRATICULE (Y OR N)";G$
70 IF G$="N" THEN INPUT "DO YOU WANT
   BORDER TICK MARKS (Y OR N)";T$
80 OPEN "I",#1,N$
90 PRINT#-2, "SC0,255,0,191;SP2;"
110 PMODE 4,1:PCLS:SCREEN 1,0
120 IF G$="N" THEN 3000
230 FOR H=0 TO 271 STEP 16
240 LINE (H,0)-(H,191),PSET
250 PRINT#-2, "PA"H",0,PD,"H",191,PU;"
255 IF INKEY$="R" THEN 5000
260 NEXT H
270 PRINT#-2, "PA128,0;"
280 FOR T=0 TO 191 STEP 4
290 PRINT#-2, "PA128,"T";YT;"
295 IF INKEY$="R" THEN 5000
300 NEXT T
310 FOR Q=1 TO 1000:NEXT Q
320 FOR V=0 TO 207 STEP 16
330 LINE (0,V)-(255,V),PSET
340 PRINT#-2, "PA0,"V",PD,255,"V",PU;"
345 IF INKEY$="R" THEN 5000
350 NEXT V
360 PRINT#-2, "PA0,96;"
370 FOR T=0 TO 255 STEP 4
380 PRINT#-2, "PA"T",96;XT;"
385 IF INKEY$="R" THEN 5000
390 NEXT T
400 FOR Q=1 TO 1000:NEXT Q
420 FOR T=0 TO 191 STEP 4
430 PSET (127,T):PSET (129,T)
440 NEXT T
450 FOR T=0 TO 255 STEP 4
460 PSET (T,95):PSET (T,97)
470 NEXT T
480 PRINT #-2, "SP1;"
500 FOR X=0 TO 255 STEP 4
600 INPUT #1, Y
700 PSET (X,Y)
720 PRINT#-2, "PA"X,Y";PD;"
750 IF INKEY$="R" THEN 5000
800 NEXT X
900 PRINT#-2, "PU;SP2;"
910 PRINT#-2, "DR-1,0;SR5,-7;"
920 PRINT#-2, "PA0,220;LB"N$;CHR$(3)
925 PRINT #-2, "SR2.5,-4
927 INPUT#1, SD
930 PRINT #-2, "PA0,250;LB0.4 VOLT/DIV
    "SD"SEC/DIV"CHR$(3)
950 PRINT#-2, "PU;SP;"
990 CLOSE #1
1000 IF INKEY$="R" THEN 20 ELSE 1000
3000 PRINT#-2, "PA0,0,PD,0,191,
     255,191,255,0,0,0,PU;"
3100 IF T$="N" THEN 500
3200 FOR T=0 TO 191 STEP 16
3300 PRINT#-2, "PA0,"T";TL1,0;YT;"
3350 IF INKEY$="R" THEN 5000
3400 NEXT T
3500 FOR D=1 TO 1000:NEXT D
3600 FOR T=0 TO 191 STEP 16
3700 PRINT#-2, "PA255,"T";TL0,1;YT"
3750 IF INKEY$="R" THEN 5000
3800 NEXT T
3900 FOR D=1 TO 1000:NEXT D
4000 FOR T=0 TO 255 STEP 16
4100 PRINT#-2, "PA"T",0;TL1,0;XT;"
4150 IF INKEY$="R" THEN 5000
4200 NEXT T
4300 FOR D=1 TO 1000:NEXT D
4400 FOR T=0 TO 255 STEP 16
4500 PRINT#-2, "PA"T",191;TL0,1;XT;"
4550 IF INKEY$="R" THEN 5000
4600 NEXT T
4700 FOR D=1 TO 1000:NEXT D
4800 GOTO 500
5000 CLOSE #1:GOTO 20
```

Storage Oscilloscope
(Continued from page 66)

the sweep subroutine, y-axis (vertical) trace data is written into the data file if D$ is Y. The sweep delay per division (SD) is saved after the trace has been saved (line 4500). This allows a suitable plotter program to correctly annotate a hard copy version of the retrieved trace. (Omit line 4500 if you don't plan to use the plotter option described below.) Finally, the disk file is closed (line 4600) after the sweep is completed and saved.

Traces which have been saved on disk can be retrieved and displayed with *or* without a graticule. This function is accomplished by the routine in lines 1000-1400. Line 1020 requests the name of the trace (N$) and line 1030 opens an input buffer (#1) for the disk. The remainder of the routine retrieves data from the disk and paints the trace on the monitor's screen. The retrieved trace is drawn using the sweep subroutine's method for painting a real-time trace. (See lines 4300-4350 and 1310-1340.) Saved traces can also be drawn on a plotter using the Scopeplt program described later.

How to Use Comscope. After loading the program (and saving it on disk if you have a disk drive), verify operation by moving the joystick handle during the display sweep phase. If the trace does not move up and down in response to joystick movements, check to make sure that the program was entered correctly.

Be sure to check all the menu options for proper operation. A displayed trace can be exited (while it's being painted or after it's frozen on the monitor) by pressing M. The screen will immediately clear and the main menu will be displayed. You can exit at other points in the program by pressing BREAK and restart by typing RUN.

Many different signal sources can be connected to the Color Computer's joystick ports as long as their output levels fall within the 0-5-volt "window." When no external amplifier is used, Comscope provides a sensitivity of about 0.4 volt per vertical division.

Low-level signals will require amplification and this is easily accomplished with a simple operational amplifier circuit such as the one described in this month's "The Electronics Scientist" column. This column also describes how to gain access to the Color Computer's joystick ports by soldering a pair of wires to terminals inside a joystick.

Three simple circuits whose outputs can be monitored with the Comscope program are shown in Fig. 2. Figure 2A shows an RC circuit that outputs a rapidly rising voltage followed by a slowly falling voltage as *C1* is charged through *R1* and then self-discharged through natural leakage paths within the circuit.

A photograph of the actual trace of the charge-discharge cycle as displayed on the computer's monitor (0.4 volt per vertical division and 0.35 second per horizontal division) is shown in Fig. 3. The trace begins automatically as soon as *S1* is closed. The switch was opened when the charge of *C1* approached maximum amplitude.

The circuit in Fig.2B is a very simple, yet ultra-sensitive, light sensor. The cadmium sulfide photoresistor, *PC1*, has a high dark resistance and a low light resistance (Radio Shack 276-116 or similar). The circuit's output voltage rises as the light level at the photoresistor increases.

Even economical printers have higher resolution than video monitors

A photograph of the monitor's screen showing two spikes produced by the light sensor when a flashlight was blinked twice in quick succession (0.4 volt per vertical division) is shown in Fig. 4. The spikes, which are easier to discern without the graticule, exhibit the very slow fall times characteristic of photoresistors. This is the so-called *memory effect*. While a small part of the delay is caused by the thermal lag of the lamp filament, note that the output from the sensor is still 0.4 volt when the second flash arrives about 2.5 seconds after the first.

A linear-response light sensor using a silicon solar cell and an op amp is shown in Fig. 2C. Using the values of *R1* and *R2* as shown, the gain of the op amp is 10. This can be increased (decreased) by increasing (decreasing) the value of *R2*.

A photograph of the monitor's screen showing the trace produced when the beam from a flashlight is slowly scanned past the solar cell is shown in Fig.5. The horizontal sweep is 1 second per division. Thanks to the X10 gain of the amplifier, the vertical scale is a sensitive 0.04 volt per division.

Earlier it was explained that traces saved on disk must have reduced resolution to avoid interruptions from the disk drive when the output buffer is filled. The display photograph in Fig.3 clearly shows an artifact resultant from a disk-drive interruption near the right end of the trace. Depending upon the signal, a similar artifact may appear at the same location on other traces.

Modifying Comscope. In using Comscope for serious applications, you may want to consider modifying or expanding the program to include custom features. Since the time per horizontal division is only approximate, you can refine the timing calibration. One possibility is to use the TIMER function.

If you can program in assembly language, you can greatly increase Comscope's sweep speed. This is done by substituting an assembly language subroutine for the sweep subroutine.

You may also wish to add sound effects, particularly if you intend to record traces over a period of minutes or hours. A simple BASIC line such as SOUND 200,50 appropriately placed will generate a tone to notify that the trace is complete or any other desired event has occured.

Using an X-Y Plotter. It's possible to use some dot-matrix printers to produce a hard-copy version of graphics displayed on a computer's monitor. This procedure is called a *screen dump*.

A better approach is to employ an x-y plotter. Even economical plotters have much higher resolution than the best available video monitors. This means very precise graticules and calibration marks can be superimposed on a plotted trace.

A program called Scopeplt that converts Comscope traces saved on disk into publication-quality hard copy is given in Listing 2. The program is designed for use with Hewlett-Packard's HP-7470A plotter and a Color Computer. With minor modifications it can also be used with other computers and any of the various plotters that understand HPGL (Hewlett-Packard Graphics Language).